

Technical Document

Document No: 11-02001
Document Title: SQL0305 Indicator variable required
Category: Hints, Tips & FAQ
Functional Area: SQL
OS/400 Release:

Document Description:

Recently I was writing a program using embedded SQL to access data from the SYSCOLUMNS table in QSYS2, using the following pieces of code:

```
C/Exec SQL
C+ Declare csrSysColumns Cursor for
C+ Select Sys_Tname, ColNo, Sys_Cname, LabelText, ColType, Length,
C+      Scale, Label, Storage
C+   From SysColumns
C+   Where Sys_Dname = :PFLIBN
C+     and Sys_Tname = :PFPPIL
C/End-Exec
```

and

```
C/Exec SQL
C+ Fetch Next from csrSysColumns
C+   into :wrkFile, :wrkFieldNo, :wrkFieldName :wrkDesc :wrkColType,
C+       :wrkLength, :wrkScale, :wrkColHeads, :wrkStorage
C/End-Exec
```

When I tested the program I noticed a couple of things. The first was that the records I had expected to select and process had not been retrieved, and the second that the joblog contained a number of diagnostic messages “**Indicator or variable required**”

The diagnostic message has message ID SQL0305 and the full details of the message are shown in the screen shot below. The message is generated whenever an SQL operation results in a null value being returned and the program is not set up to deal with it.

In my case it was the **wrkDesc** field that was having a null value returned to it and the program hadn’t been coded to handle with the situation.

```

Additional Message Information

Message ID . . . . . :   SQL0305           Severity . . . . . :   30
Message type . . . . . :   Diagnostic
Date sent . . . . . :   28/09/04           Time sent . . . . . :   16:53:13

Message . . . . . :   Indicator variable required.
Cause . . . . . :   A FETCH, an embedded SELECT, a CALL or a SET or VALUES
INTO statement has resulted in a null value, but an indicator variable was
not specified for host variable WRKDESC. The relative position of the host
variable in the INTO clause or parameter list is 4. If the host variable
name is *N, an SQLDA was specified.
Recovery . . . . . :   Specify an indicator variable, and precompile the program
again.

                                                                    Bottom

Press Enter to continue.

F3=Exit   F6=Print   F9=Display message details   F12=Cancel
F21=Select assistance level

```

The solution is to use an indicator variable alongside the fields you are retrieving from the database file. This indicator variable is then set by SQL whenever a null value is returned for the associated field. An indicator field is defined as a two byte Binary field (that is, data type “B” and not data type “I” as you might expect if you’re familiar with API programming) and is entered on the **FETCH** instruction alongside it’s associated host variable:

```

D indDesc           s           2B 0
D indColType        s           2B 0
D indLength         s           2B 0
D indScale          s           2B 0
D indColHeads       s           2B 0
D indStorage        s           2B 0

C/Exec SQL
C+ Fetch Next from csrSysColumns
C+  into :wrkFile,
C+      :wrkFieldNo,
C+      :wrkFieldName,
C+      :wrkDesc      :indDesc,
C+      :wrkColType   :indColType,
C+      :wrkLength    :indLength,
C+      :wrkScale     :indScale,
C+      :wrkColHeads  :indColHeads,
C+      :wrkStorage   :indStorage
C/End-Exec

```

I have spread the variables over a number of lines to make the command easier to read. Note, that a) not all variables have to have indicator variables assigned, and b) that there is no separator comma between the host variable and its associated indicator variable. The separator comma comes after the indicator variable.

When the record containing null data in the **wrkDesc** field is now returned, the **indDesc** indicator variable is set with a negative value. For indicator variables, a negative value indicates that the host variable is null, whilst a non-negative value (zero or above) indicates that the host variable is not null.

** End of Document **